DIGITAL ICS DESIGN

DSP48A1 PROJECT

PREPARED BY: ABDALRAHMAN ALI ELTAHER
Submitted to: Eng. Kareem Waseem & Eng. Mai Sherif

❖ Design Diagram



❖ The FPGA utilized in the project



RTL code :

- *MUX_4*

```
module mux_4(i_sel,i_in0,i_in1,i_in2,i_in3,o_out);

    //inputs
    input [1:0] i_sel;
    input [47:0] i_in0,i_in1,i_in2,i_in3;

    //output
    output reg [47:0] o_out;

    always @(*) begin
        case (i_sel)
            2'b00 : o_out = i_in0;
            2'b01 : o_out = i_in1;
            2'b10 : o_out = i_in2;
            2'b11 : o_out = i_in3;
            default o_out = 48'd0;
        endcase
    end

endmodule
```

- DFF *with MUX*

```verilog
module dff_mux(i_clk,i_rst,i_en,i_data,o_data);

    //parameter
    parameter WIDTH = 18;
    parameter SEL = 1;
    parameter RSTTYPE = "SYNC"; //ASYNC or SYNC

    //inputs
    input i_en,i_clk,i_rst;
    input [WIDTH-1:0] i_data;

    //output
    output [WIDTH-1:0] o_data;

    reg [WIDTH-1:0] q;

    generate
        if (RSTTYPE == "SYNC") begin
            always @(posedge i_clk) begin
                if(i_rst)
                    q <= 0;
                else begin
                    if(i_en)
                        q <= i_data;
                end
            end
        end
        else if(RSTTYPE == "ASYNC") begin
            always @(posedge i_clk or posedge i_rst) begin
                if(i_rst)
                    q <= 0;
                else begin
                    if(i_en)
                        q <= i_data;
                end
            end
        end
    endgenerate

    assign o_data =(SEL == 1)? q:i_data;

endmodule
```

➢ *DSP48A1*

- *Parameters,Inputs,Outpus,wires*

```
module DSP48A1(A,B,C,D,CLK,CARRYIN,OPMODE,BCIN,
                RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE,
                    CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,
                        CEOPMODE,PCIN,BCOUT,PCOUT,P,M,CARRYOUT,CARRYOUTF);

    //========================== PARAMETERS ==========================
    parameter A0REG = 0; parameter A1REG = 1;
    parameter B0REG = 0; parameter B1REG = 1;
    parameter CREG = 1;
    parameter DREG = 1;
    parameter MREG = 1;
    parameter PREG = 1;
    parameter CARRYINREG = 1;
    parameter CARRYOUTREG = 1;
    parameter OPMODEREG = 1;
    parameter CARRYINSEL = "OPMODE5";//OPMODE5 OR CARRYIN
    parameter B_INPUT = "DIRECT"; //DIRECT OR CASCADE
    parameter RSTTYPE = "SYNC"; //SYNC OR ASYNC
    //==================================================================

    //========================== INPUTS ==========================
    input CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,CEOPMODE;
    input RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE;
    input CLK;
    input CARRYIN;
    input [7:0]OPMODE;
    input [17:0] A,B,D,BCIN;
    input [47:0] C,PCIN;
    //==============================================================

    //========================== OUTPUTS ==========================
    output CARRYOUT,CARRYOUTF;
    output [17:0] BCOUT;
    output [35:0] M;
    output [47:0] P,PCOUT;
    //==============================================================

    //========================== WIRES ==========================
    wire [17:0] A0_REG,A1_REG,B0_REG,B1_REG,D_REG,B_out;
    wire [47:0] C_REG,post_add_sub,out_X,out_Z;
    wire [35:0] M_REG,out_multiplier;
    wire [17:0] pre_add_sub,i_B1_REG;
    wire i_CYI,o_CYI;
    wire i_CYO;
    wire o_opmode_4,o_opmode_5,o_opmode_6,o_opmode_7;
    wire [1:0] o_opmode_0_1,o_opmode_3_2;
    //==============================================================
```

- *Instantiation DFF with MUX*

```
//================================================================ INSTANTIATIONS DFF_MUX ================================================================
dff_mux #(.WIDTH(18),.SEL(A0REG),.RSTTYPE(RSTTYPE)) a0reg(.i_clk(CLK),.i_rst(RSTA),.i_en(CEA),.i_data(A),.o_data(A0_REG));
dff_mux #(.WIDTH(18),.SEL(A1REG),.RSTTYPE(RSTTYPE)) a1reg(.i_clk(CLK),.i_rst(RSTA),.i_en(CEA),.i_data(A0_REG),.o_data(A1_REG));
dff_mux #(.WIDTH(18),.SEL(B0REG),.RSTTYPE(RSTTYPE)) b0reg(.i_clk(CLK),.i_rst(RSTB),.i_en(CEB),.i_data(B_out),.o_data(B0_REG));
dff_mux #(.WIDTH(18),.SEL(B1REG),.RSTTYPE(RSTTYPE)) b1reg(.i_clk(CLK),.i_rst(RSTB),.i_en(CEB),.i_data(i_B1_REG),.o_data(B1_REG));
dff_mux #(.WIDTH(18),.SEL(DREG),.RSTTYPE(RSTTYPE)) dreg(.i_clk(CLK),.i_rst(RSTD),.i_en(CED),.i_data(D),.o_data(D_REG));
dff_mux #(.WIDTH(48),.SEL(CREG),.RSTTYPE(RSTTYPE)) creg(.i_clk(CLK),.i_rst(RSTC),.i_en(CEC),.i_data(C),.o_data(C_REG));
dff_mux #(.WIDTH(36),.SEL(MREG),.RSTTYPE(RSTTYPE)) mreg(.i_clk(CLK),.i_rst(RSTM),.i_en(CEM),.i_data(out_multiplier),.o_data(M_REG));
dff_mux #(.WIDTH(1),.SEL(CARRYINREG),.RSTTYPE(RSTTYPE))  CYI(.i_clk(CLK),.i_rst(RSTCARRYIN),.i_en(CECARRYIN),.i_data(i_CYI),.o_data(o_CYI));
dff_mux #(.WIDTH(1),.SEL(CARRYOUTREG),.RSTTYPE(RSTTYPE)) CYO(.i_clk(CLK),.i_rst(RSTCARRYIN),.i_en(CECARRYIN),.i_data(i_CYO),.o_data(CARRYOUT));
dff_mux #(.WIDTH(48),.SEL(PREG),.RSTTYPE(RSTTYPE)) preg(.i_clk(CLK),.i_rst(RSTP),.i_en(CEP),.i_data(post_add_sub),.o_data(P));

dff_mux #(.WIDTH(2),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode01(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[1:0]),.o_data(o_opmode_0_1));
dff_mux #(.WIDTH(2),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode32(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[3:2]),.o_data(o_opmode_3_2));
dff_mux #(.WIDTH(1),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode4(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[4]),.o_data(o_opmode_4));
dff_mux #(.WIDTH(1),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode5(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[5]),.o_data(o_opmode_5));
dff_mux #(.WIDTH(1),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode6(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[6]),.o_data(o_opmode_6));
dff_mux #(.WIDTH(1),.SEL(OPMODEREG),.RSTTYPE(RSTTYPE)) opmode7(.i_clk(CLK),.i_rst(RSTOPMODE),.i_en(CEOPMODE),.i_data(OPMODE[7]),.o_data(o_opmode_7));
//========================================================================================================================================================
```

- *Instantiation MUX_4*

```
//================================================ INSTAANTIATION MUX_4 ================================================
mux_4 X(.i_sel(o_opmode_0_1),.i_in0(48'd0),.i_in1({12'd0,M_REG}),.i_in2(P),.i_in3({D[11:0],A1_REG,B1_REG}),.o_out(out_X));
mux_4 Z(.i_sel(o_opmode_3_2),.i_in0(48'd0),.i_in1(PCIN),.i_in2(P),.i_in3(C_REG),.o_out(out_Z));
//======================================================================================================================
```

- *Assign*

```
    //========================================= assign =========================================
    assign B_out = (B_INPUT == "DIRECT")?B:(B_INPUT == " CASCADE")?BCIN:0;
    assign pre_add_sub = (o_opmode_6)?(D_REG-B0_REG) : (D_REG+B0_REG) ;
    assign i_B1_REG = (o_opmode_4)?pre_add_sub:B0_REG;
    assign out_multiplier = B1_REG * A1_REG;
    assign i_CYI = (CARRYINSEL == "OPMODE5")?o_opmode_5:(CARRYINSEL == "CARRYIN")?CARRYIN:0;
    assign {i_CYO,post_add_sub} = (o_opmode_7)?(out_Z-(out_X+o_CYI)):(out_X+out_Z+o_CYI);
    assign BCOUT = B1_REG;
    assign M = M_REG;
    assign CARRYOUTF = CARRYOUT;
    assign PCOUT = P;
    //==========================================================================================

endmodule
```

❖ Testbench code

- *Inputs and Outpus*

```
module DSP48A1_tb();

    //========================== INPUTS ==========================
    reg CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,CEOPMODE;
    reg RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE;
    reg CLK;
    reg CARRYIN;
    reg [7:0]OPMODE;
    reg [17:0] A,B,D,BCIN;
    reg [47:0] C,PCIN;
    //============================================================

    //========================== OUTPUTS ==========================
    wire CARRYOUT,CARRYOUTF;
    wire [17:0] BCOUT;
    wire [35:0] M;
    wire [47:0] P,PCOUT;
    //============================================================
```

- *Instantiation DUT*

```
// ==================================================== Instantiation DUT ====================================================
DSP48A1 #(.A0REG(0),.A1REG(1),.B0REG(0),.B1REG(1),.CREG(1),.DREG(1),.MREG(1),.PREG(1),.CARRYINREG(1),
            .CARRYOUTREG(1),.OPMODEREG(1),.CARRYINSEL("OPMODE5"),.B_INPUT("DIRECT"),.RSTTYPE("SYNC"))
                DUT(A,B,C,D,CLK,CARRYIN,OPMODE,BCIN,
                    RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE,
                        CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,
                            CEOPMODE,PCIN,BCOUT,PCOUT,P,M,CARRYOUT,CARRYOUTF);
// ==========================================================================================================================
```

- *Clock generation*

```
// ========= clock generation =========
initial begin
    CLK = 0;
    forever #1 CLK = ~CLK;
end
// ====================================
```

- *Stimulus generations*

  2.1 Verify Reset Operation

```
// ============================================ Stimulus Generation ============================================
initial begin
    //2.1 Verify Reset Operation
    {RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE} = {8'b1111_1111};
    {CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,CEOPMODE} = {8'b0101_0010};
    CARRYIN = $random;
    OPMODE = $random;
    A = $random;
    B = $random;
    D = $random;
    BCIN = $random;
    C = $random;
    PCIN = $random;
    @(negedge CLK);
    if (CARRYOUT != 1'b0 || CARRYOUTF != 1'b0 || BCOUT !=18'd0 || M != 36'd0 || P !=48'd0 || PCOUT != 48'd0) begin
        $display( "Time=%0t,CARRYOUT =%b,CARRYOUTF =%b,BCOUT =%h,M =%b,P=%h,PCOUT=%h", $time,CARRYOUT,CARRYOUTF,BCOUT,M,P,PCOUT);
        $stop;
    end
    {RSTA,RSTB,RSTM,RSTP,RSTC,RSTD,RSTCARRYIN,RSTOPMODE} = {8'b0000_0000};
    {CEA,CEB,CEM,CEP,CEC,CED,CECARRYIN,CEOPMODE} = {8'b1111_1111};
```

2.2. Verify DSP Path 1



Figure 1: Path 1 Data Flow

```
//2.2. Verify DSP Path 1
OPMODE = 8'b1101_1101;
A = 18'd20; B = 18'd10; C = 48'd350 ; D = 18'd25;
BCIN = $random;
CARRYIN = $random;
PCIN = $random;
repeat(4) @(negedge CLK);
if (CARRYOUT != 1'b0 || CARRYOUTF != 1'b0 || BCOUT !=18'hf || M != 36'h12c || P !=48'h32 || PCOUT != 48'h32) begin
    $display( "Time=%0t,CARRYOUT =%b,CARRYOUTF =%b,BCOUT =%h,M =%h,P=%h,PCOUT=%h", $time,CARRYOUT,CARRYOUTF,BCOUT,M,P,PCOUT);
    $stop;
end
```

2.3. Verify DSP Path 2



Figure 2: Path 2 Data Flow

```
//2.3. Verify DSP Path 2
OPMODE = 8'b0001_0000;
A = 18'd20; B = 18'd10; C = 48'd350 ; D = 18'd25;
BCIN = $random;
CARRYIN = $random;
PCIN = $random;
repeat(3) @(negedge CLK);
if (CARRYOUT != 1'b0 || CARRYOUTF != 1'b0 || BCOUT !=18'h23 || M != 36'h2bc || P !=48'd0 || PCOUT != 48'd0) begin
    $display( "Time=%0t,CARRYOUT =%b,CARRYOUTF =%b,BCOUT =%h,M =%h,P=%h,PCOUT=%h", $time,CARRYOUT,CARRYOUTF,BCOUT,M,P,PCOUT);
    $stop;
end
```

2.4. Verify DSP Path 3



Figure 3: Path 3 Data Flow

```verilog
//2.4. Verify DSP Path 3
OPMODE = 8'b0000_1010;
A = 18'd20; B = 18'd10; C = 48'd350 ; D = 18'd25;
BCIN = $random;
CARRYIN = $random;
PCIN = $random;
repeat(3) @(negedge CLK);
if (CARRYOUT != 1'b0 || CARRYOUTF != 1'b0 || BCOUT !=18'ha || M != 36'hc8 || P !=48'd0 || PCOUT != 48'd0) begin
    $display( "Time=%0t,CARRYOUT =%b,CARRYOUTF =%b,BCOUT =%h,M =%h,P=%h,PCOUT=%h", $time,CARRYOUT,CARRYOUTF,BCOUT,M,P,PCOUT);
    $stop;
end
```

2.5. Verify DSP Path 4



Figure 4: Path 4 Data Flow

```
        //2.5. Verify DSP Path 4
        OPMODE = 8'b1010_0111;
        A = 18'd5; B = 18'd6; C = 48'd350 ; D = 18'd25; PCIN = 48'd3000;
        BCIN = $random;
        CARRYIN = $random;
        repeat(3) @(negedge CLK);
        if (CARRYOUT != 1'b1 || CARRYOUTF != 1'b1 || BCOUT !=18'h6 || M != 36'h1e || P !=48'hfe6fffec0bb1 || PCOUT != 48'hfe6fffec0bb1) begin
            $display( "Time=%0t,CARRYOUT =%b,CARRYOUTF =%b,BCOUT =%h,M =%h,P=%h,PCOUT=%h", $time,CARRYOUT,CARRYOUTF,BCOUT,M,P,PCOUT);
            $stop;
        end
        $stop;
    end
    // ==============================================================================================================

endmodule
```

❖ **DO FILE**

run.do

```
vlib work
vlog DSP48A1.v DSP48A1_tb.v
vsim -voptargs=+acc work.DSP48A1_tb
add wave -r *
run -all
#quit -sim
```

❖ QuestaSim Snippets





❖ Constrain File

```
set_property -dict { PACKAGE_PIN W5   IOSTANDARD LVCMOS33 } [get_ports CLK]
create_clock -add -name sys_clk_pin -period 10.000 -waveform {0 5} [get_ports CLK]
```

❖ Elaboration

➢ *Messages tab*

- Vivado Commands (3 infos, 4 status messages)
  - General Messages (3 infos, 4 status messages)
    - [IP_Flow 19-234] Refreshing IP repositories
    - [IP_Flow 19-1704] No user IP repositories specified
    - [IP_Flow 19-2313] Loaded Vivado IP repository 'C:/Xilinx/Vivado/2018.2/data/ip'.
    - Command: synth_design -rtl -name rtl_1 (3 more like this)
- Elaborated Design (2 infos, 5 status messages)
  - General Messages (2 infos, 5 status messages)
    - [Project 1-570] Preparing netlist for logic optimization
    - Processing XDC Constraints (4 more like this)
      - Initializing timing engine
      - Parsing XDC File [constrain.xdc]
      - Finished Parsing XDC File [constrain.xdc]
      - Completed Processing XDC Constraints
    - [Opt 31-138] Pushed 0 inverter(s) to 0 load pin(s).

➢ *Schematic snippets*



❖ Synthesis

➢ *Messages tab*





- *Utilization report*

Name	Slice LUTs (134600)	Slice Registers (269200)	DSPs (740)	Bonded IOB (500)	BUFGCTRL (32)
DSP48A1	231	160	1	327	1
a1reg (dff_mux__para...	0	18	0	0	0
b1reg (dff_mux__para...	0	18	0	0	0
creg (dff_mux__param...	0	48	0	0	0
CYI (dff_mux__parame...	1	1	0	0	0
CYO (dff_mux__param...	0	1	0	0	0
dreg (dff_mux__param...	18	18	0	0	0
mreg (dff_mux__para...	1	0	1	0	0
opmode4 (dff_mux__p...	0	1	0	0	0
opmode5 (dff_mux__p...	0	1	0	0	0
opmode6 (dff_mux__p...	17	1	0	0	0
opmode7 (dff_mux__p...	1	1	0	0	0
opmode01 (dff_mux__...	144	2	0	0	0
opmode32 (dff_mux__...	49	2	0	0	0
preg (dff_mux__param...	0	48	0	0	0

- *timing report*

Design Timing Summary

Setup

Worst Negative Slack (WNS):	5.925 ns
Total Negative Slack (TNS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	106

Hold

Worst Hold Slack (WHS):	0.182 ns
Total Hold Slack (THS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	106

Pulse Width

Worst Pulse Width Slack (WPWS):	4.500 ns
Total Pulse Width Negative Slack (TPWS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	162

All user specified timing constraints are met.

- *Schematic snippets*



❖ Implementation

➢ *Messages tab*





➢ *Utilization report*

Name	Slice LUTs (133800)	Slice Registers (267600)	Slice (33450)	LUT as Logic (133800)	LUT Flip Flop Pairs (133800)	DSPs (740)	Bonded IOB (500)	BUFGCTRL (32)
DSP48A1	230	179	93	230	52	1	327	1
a1reg (dff_mux__para...	0	18	8	0	0	0	0	0
b1reg (dff_mux__para...	0	36	10	0	0	0	0	0
creg (dff_mux__param...	0	48	15	0	0	0	0	0
CYI (dff_mux__parame...	1	1	1	1	1	0	0	0
CYO (dff_mux__param...	0	2	2	0	0	0	0	0
dreg (dff_mux__param...	18	18	15	18	0	0	0	0
mreg (dff_mux__para...	0	0	0	0	0	1	0	0
opmode4 (dff_mux__p...	0	1	1	0	0	0	0	0
opmode5 (dff_mux__p...	0	1	1	0	0	0	0	0
opmode6 (dff_mux__p...	17	1	6	17	0	0	0	0
opmode7 (dff_mux__p...	1	1	2	1	0	0	0	0
opmode01 (dff_mux__...	144	2	44	144	0	0	0	0
opmode32 (dff_mux__...	49	2	30	49	0	0	0	0
preg (dff_mux__param...	0	48	12	0	0	0	0	0

➢ *timing report*

Design Timing Summary

Setup

Worst Negative Slack (WNS):	4.827 ns
Total Negative Slack (TNS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	125

Hold

Worst Hold Slack (WHS):	0.139 ns
Total Hold Slack (THS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	125

Pulse Width

Worst Pulse Width Slack (WPWS):	4.500 ns
Total Pulse Width Negative Slack (TPWS):	0.000 ns
Number of Failing Endpoints:	0
Total Number of Endpoints:	181

All user specified timing constraints are met.

➢ *device snippets*